SECTION - A MICRO-COMPUTER PROGRAM
(93, 94)

By W. E. Kilby

Cross-sectional displays of geologic data are one of the mainstays of geological interpretaion, second only to map (plan) displays. Unfortunately, time intensive and error prone manual techniques required to produce all but the simplest form of cross-sections, horizontal projections onto a vertical section, have limited the usefulness of this display form. Horizontal projections are valid when the fold-axes are horizontal, but invalid if the structures have any plunge. Profiles are sections oriented normal to the projection direction; they offer one of the best ways of displaying geologic data to aid interpretation; they display true thickness versus apparent thickness.

```
1 REM************************** SECTION *************************
2 REM***                        W.E.KILBY                     ***
3 REM***********************************************************
10 PRINT"                        ***SECTION***"
20 PI=3.14159:DR=(2*PI)/360:RD=360/(2*PI)
30 INPUT"PROJECTION DIRECTION, TREND AND PLUNGE";TR,PL
40 INPUT"SECTION VIEWING DIRECTION, TREND AND PLUNGE";VT,VP
50 INPUT"SECTION ORIGIN, X,Y,Z";SX,SY,SZ
60 T1=TR:T2=PL:T3=VT:T4=VP
70 TR=TR*DR:PL=PL*DR:VT=VT*DR:VP=VP*DR:NI=PI/2
80 L1=SIN(VT+NI):M1=COS(VT+NI):N1=0
90 L2=SIN(VT)*COS(VP-NI):M2=COS(VT)*COS(VP-NI):N2=-SIN(VP-NI)
100 L3=SIN(VT)*COS(VP):M3=COS(VT)*COS(VP):N3=-SIN(VP)
110 TR=TR-VT:PL=PL-VP
119 REM INPUT DATA
120 FOR I=1 TO 1000
130 INPUT"ENTER DATA; X,Y,Z,DIP-DIR.,DIP";X,Y,Z,DD,DI
140 IF X=9 AND Y=9 AND Z=9 AND DD=9 AND DI=9 THEN END
150 IF DI>90 THEN DI=180-DI:DD=DD+180
160 X=X-SX:Y=Y-SY:Z=Z-SZ:DD=DD*DR:DI=DI*DR
169 REM CALCULATE SECTION COORDINATES
170 X1=L1*X+M1*Y+N1*Z:Y1=L2*X+M2*Y+N2*Z:Z1=L3*X+M3*Y+N3*Z
180 IF TR=0 AND PL=0 THEN 200
190 X1=X1-TAN(TR)*Z1:ZZ=1/COS(TR)*Z1:Y1=Y1+SIN(PL)*ZZ
199 REM CALCULATE PITCH
200 IF DD<0 AND DI<0 THEN PT=999:GOTO 240
210 DD=DD+PI:DI=NI-DI:L=SIN(DD)*COS(DI):M=COS(DD)*COS(DI):N=-SIN(DI)
220 LL=L*L1+M*M1+N*N1:MM=L*L2+M*M2+N*N2:PT=ATN(LL/MM)*RD
230 IF PT<0 THEN PT=180+PT
239 REM OUTPUT DATA
240 PRINT X1,Y1,Z1,PT
250 NEXT I
260 END
```



OUTPUT DATA

X	Y	Z	Pitch
6.5	1.7	−1.4	999
17.4	−4.1	3.3	999
23.7	−5.5	4.5	999
25.9	3.8	−3.1	999
27.5	15.6	−12.6	999
11.6	3.6	−2.9	999
20.6	0.4	−0.3	999
23.7	3.2	−2.6	999
25.6	18.1	−14.7	999
7.8	0.0	0.0	11.0
21.4	−7.6	6.2	180
23.0	4.1	−3.3	147.8
27.6	8.4	−6.8	114.9

SECTION A–A'
LOOKING 320/–25



OUTPUT DATA

X	Y	Z	PITCH
8.5	−1.7	3	999
22.7	−7.7	14	999
30.9	−10.5	19	999
33.7	−7.7	14	999
35.9	−3.9	7	999
15.2	−2.8	5	999
26.9	−7.2	13	999
30.9	−7.2	13	999
33.3	−2.2	4	999
10.2	−2.8	5	24.7
27.9	−10.5	19	16.5
30.1	−6.6	12	167.0
36.1	−6.6	12	123.3

Figure 104. Cross-section plots of data projected onto two different planes of section. (a) profile A-A', constructed normal to the projection direction (fold-axis), (b) vertical east-west section A-B'. Note distortion of structure in (b) due to oblique angle between structure orientation and section orientation.

The program described here, SECTION (Table 1), calculates the position of a data point on a section of any orientation, after being projected parallel to any desired orientation; and calculates the apparent dip of any orientation data on the plane of the section. The program is written in BASIC with no machine specific features so it can be entered and run on virtually any micro-computer. The user specifies map location and elevation of the map or drill hole data, the orientation and position of the section, and the projection direction. The program returns the resultant section coordinates and the apparent dip of any planar features (Fig. 104 a and b).

SECTION POSITION - It is necessary to position the section relative to the data being investigated. The X, Y, Z (easting, northing, elevation) coordinates specified for the section position become the origin position on the cross-section plot. The specified X and Y map position becomes the zero horizontal position and the Z map coordinate value becomes the zero vertical position on the section plot.

SECTION ORIENTATION - An infinite variety of section orientations can be constructed through any given position. The section orientation is specified by giving the trend and plunge of the direction in which the user wishes to look at the data. For example, a viewing direction of 0/0 (north and horizontal) would result in a vertical section oriented east-west. The viewing direction is perpendicular to the plane of section.

PROJECTION DIRECTION - All data is projected parallel to the given orientation onto the plane of section. A variety of graphical and statistical techniques have been described to aid in the determination of the best-fit fold-axis orientation (Charlesworth, *et al.*, 1976).

INPUT DATA - Positional data is entered as X, Y, Z map coordinates (Fig. 103). These coordinates can be in any map units measured on any grid but the grid must be orthogonal and have the same scale in all directions. Planar orientation data is entered in the form of dip-direction and dip; dip-direction is the clockwise angle measured from grid north. Dip is the vertical downward angle from horizontal to the surface of the structure measured parallel to the dip-direction orientation; dip angles greater than 90 degrees denote overturned bedding. Linear feature orientations should be entered as trend and plunge measurements. Enter a negative (-) value for the dip-direction and dip if no orientation is available for an entry, a pitch value of 999 will be returned to indicate no orientation data was available.

OUTPUT DATA - The position of data on the plane of section is returned as X, Y, Z coordinates, that is the horizontal and vertical distances from the origin in the section, and the normal distance of the point from the plane of section. The apparent dip of orientation data on the section is reported as a clockwise pitch angle.

To illustrate the use of SECTION a hypothetical set of data is employed. Figure 103 contains the raw data and a map representation of this data. In this example the data is projected parallel to the hypothetical fold-axis, 140/25 (trend/plunge). Two cross-sections are constructed; the first, A-A' is a profile, which is oriented normal to the fold-axis (Fig. 104a), the second A-B' is oriented vertical and east-west (Fig. 104b). The results in graphic and numeric form for these two section orientations are given on the figure; this data can be used to check the accuracy of program entry. Figure 105 is a representation of the screen display during program execution.



Figure 105. Example of screen display during program execution. User input is underlined.

The program as given is short, about 1 000 bytes without the REM statements, but extremely powerful. Modifications to the program to allow file handling and plotter output turn this program into a very sophisticated geological tool. The author has used this program in conjunction with larger programs which produce such varied products as isometric net diagrams of grid surfaces (*see* Grieve and Kilby, this volume), borehole deviation diagrams from dip-metre surveys, rotation of map data from one grid system to another, and varied forms of cross-section displays.

REFERENCES

Charlesworth, H.A.K. and Kilby, W. E. (1981): Calculating Thickness from Outcrop and Drill-hole Data, *Cdn. Pet. Geol.*, Bull., Vol. 29, No. 2, pp. 277-292.

Charlesworth, H.A.K., Langenberg, C. W., and Ramsden, J. (1976): Determining Axes, Axial Planes and Sections of Macroscopic Folds Using Computer-based Methods, *Cdn. Jour. Earth Sci.*, Vol. 13, pp. 54-65.



Figure 106. Sustut coal licences, property location map.